Pensieve header: Mathematica notebook accompanying Talks: LesDiablerets-2601.

Initialization

```
In[∘]:= SetDirectory["C:\\drorbn\\AcademicPensieve\\Talks\\LesDiablerets-2601"];
Once[<< KnotTheory`]
SetOptions[$FrontEndSession, PrintingStyleEnvironment → "Working"];
```

```
In[∘]:= PD[epd_EPD] := PD @@ epd /. {X_{i_,j_} :→ X[j, i + 1, j + 1, i], X̄_{i_,j_} :→ X[j, i, j + 1, i + 1]}
```

```
In[∘]:= Rot[{X_, φ_}] := {X, φ};
Rot[pd_PD] := Module[{n, xs, x, rots, Xp, Xm, front = {1}, k},
    n = Length@pd; rots = Table[0, {2 n + 1}];
    xs = Cases[pd, x_X :→ { Xp[x[[4]], x[[1]]]  PositiveQ@x
                            Xm[x[[2]], x[[1]]]  True      ];
    For[k = 1, k ≤ 2 n, ++k,
     If[FreeQ[front, -k],
      front = Flatten@Replace[front, k → (xs /. {
              Xp[k, l_] | Xm[l_, k] :→ {l + 1, k + 1, -l},
              Xp[l_, k] | Xm[k, l_] :→ (++rots[[l]]; {-l, k + 1, l + 1}),
              _Xp | _Xm :→ {}
             }), {1}],
      Cases[front, k | -k] /. {k, -k} :→ --rots[[k]];
     ]
    ];
    {xs /. {Xp[i_, j_] :→ {+1, i, j}, Xm[i_, j_] :→ {-1, i, j}}, rots} ];
Rot[K_] := Rot[PD[K]];
```

```
In[∘]:= Options[PolyPlot1] = {Labeled → False};
PolyPlot1[Δ_, OptionsPattern[]] := Module[{Δ1, crs, m, maxc, minc, s, rect},
   Δ1 = PowerExpand@Expand@Δ;
   rect = {{0, 0}, {1, 0}, {1, 1}, {0, 1}};
   If[Δ1 === 0, Graphics[],
    m = Max[-Exponent[Δ1, T, Min], Exponent[Δ1, T, Max]];
    crs = CoefficientRules[T^m Δ1, {T}];
    maxc = N@Log@Max@Abs[Last /@ crs];
    minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
    If[minc == maxc, s[_] = 0, s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
    Graphics[crs /. ({x_} → c_) :→ {
        Lighter[Which[c == 0, White, c > 0, Red, c < 0, Blue], 0.88 s[Abs@c]],
        Tooltip[Polygon[({x + m - 1 / 2, 0} + #) & /@ rect], c T^(x-m)],
        If[Not@OptionValue[Labeled], {},
```

```
      {Black, FontSize → Scaled[1/((m + 1) (6 + maxc))], Text[c T^(x-m), {x + m, 0.5}]}]
    }, AspectRatio → Min[1/5, 1/Sqrt[m + 1]],
   ImagePadding → None, PlotRangePadding → None]
 ]];
Options[PolyPlot2] = {Labeled → False, Shear → True};
PolyPlot2[θ_, OptionsPattern[]] :=
 Module[{θ1, crs, m1, m2, maxc, minc, s, stone, mat, p},
  θ1 = PowerExpand@Expand@θ;
  If[θ1 === 0, Graphics[{White, Disk[]}],
   If[OptionValue[Shear],
    stone = Table[{Cos[α], Sin[α]} / Cos[2 π / 12] / 2, {α, 2 π / 12, 2 π, 2 π / 6}];
    mat = {{1, -1/2}, {0, Sqrt[3]/2}},
    stone = {{1, 1}, {-1, 1}, {-1, -1}, {1, -1}} / 2;
    mat = IdentityMatrix[2];
   ];
   m1 = Max[-Exponent[θ1, T1, Min], Exponent[θ1, T1, Max]];
   m2 = Max[-Exponent[θ1, T2, Min], Exponent[θ1, T2, Max]];
   crs = CoefficientRules[T1^m1 T2^m2 θ1, {T1, T2}];
   maxc = N@Log@Max@Abs[Last /@ crs];
   minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
   If[minc == maxc, s[_] = 0, s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
   Graphics[{ (*{Yellow,Disk[{0,0},1+Cos[2π/12]Norm[{m1,m2}]/Sqrt[2]]},*)
     crs /. ({x1_, x2_} → c_) :> {
        Lighter[Which[c == 0, White, c > 0, Red, c < 0, Blue], 0.88 s[Abs@c]],
        p = mat.{x1 - m1, x2 - m2};
        Tooltip[Polygon[(p + #) & /@ stone], c T1^(x1-m1) T2^(x2-m2)],
        If[Not@OptionValue[Labeled], {},
         {Black, FontSize → Scaled[1/(Max[m1, m2] (10 + maxc))], Text[c T1^(x1-m1) T2^(x2-m2), p]}]
       }
     }, ImagePadding → None, PlotRangePadding → None]
  ]];
Options[PolyPlot] = {Labeled → False, ImageSize → Automatic};
PolyPlot[{Δ_, θ_}, opts___Rule] := GraphicsColumn[
  {PolyPlot1[Δ, FilterRules[Join[{opts}, Options[PolyPlot]], Options[PolyPlot1]]],
   PolyPlot2[θ, FilterRules[Join[{opts}, Options[PolyPlot]], Options[PolyPlot2]]]},
  Spacings → Scaled@0.08, ImagePadding → None, PlotRangePadding → None,
  FilterRules[Join[{opts}, Options[PolyPlot]], Options[GraphicsColumn]]
 ];
```

PPDemo

In[∘]:= **PPDemoRect** =
PolyPlot$\left[\left\{2\,T^{-1} - 1 + 3\,T,\ 2 + T_1 - T_1\,T_2 + T_2 - T_1^{-1} + T_1^{-1}\,T_2^{-1} - T_2^{-1}\right\},\ \mathbf{Labeled} \to \mathbf{True},\ \mathbf{Shear} \to \mathbf{False}\right]$

Out[∘]=



In[∘]:= **Export**["figs/PPDemoRect.pdf", **PPDemoRect**]

Out[∘]=

figs/PPDemoRect.pdf

In[•]:=
```
PPDemoHex = PolyPlot[{2 T^-1 - 1 + 3 T, 2 + T1 - T1 T2 + T2 - T1^-1 + T1^-1 T2^-1 - T2^-1}, Labeled → True]
```
Out[•]=



In[•]:=
```
Export["figs/PPDemoHex.pdf", PPDemoHex]
```
Out[•]=
```
figs/PPDemoHex.pdf
```

Beehive

In[•]:=
```
Top18Data =
  (Get["C:\\drorbn\\AcademicPensieve\\Projects\\HigherRank\\DunfieldKnots\\D" <> ToString[
        #] <> ".m"][[2, 2]] /. {T1 → T1, T2 → T2}) & /@
   {297, 298, 299, 300, 301, 302, 303, 304, 305, 306, 307, 308, 309, 310, 311, 312, 313, 317};
```
In[•]:=
```
Top18Images = PolyPlot2 /@ Top18Data;
```

```
Show[Top18 = Graphics[{
    {0, 0, 1}, {1, 0, 2}, {2, 0, 3},
    {0, 1, 4}, {1, 1, 5},
    {-1, 2, 6}, {0, 2, 7}, {1, 2, 8},
    {-1, 3, 9}, {0, 3, 10},
    {-2, 4, 11}, {-1, 4, 12}, {0, 4, 13},
    {-2, 5, 14}, {-1, 5, 15},
    {-3, 6, 16}, {-2, 6, 17}, {-1, 6, 18}
   },
   PlotRange → {{-0.6, 6.1}, {-0.5, 2.5}}, ImageSize → 1200] /. {x_, y_, k_} ⧴
   Inset[
    SetAlphaChannel[Top18Images⟦k⟧, Graphics[{White,
       Polygon[Table[{Cos[t], Sin[t]}, {t, 0, 5 π / 3, π / 3}]]}, Background → Black]],
    x {0, 1} + y {√3 / 2, 1 / 2}, Center, 1.2
   ], ImageSize → Small]
```

In[∘]:=
```
Export["Top18.pdf", ImageCrop[Top18]]
```

In[∘]:=
```
Data51to68 =
  (Get["C:\\drorbn\\AcademicPensieve\\Projects\\HigherRank\\DunfieldKnots\\D0" <>
       ToString[#] <> ".m"]⟦2, 2⟧ /. {T1 → T_1, T2 → T_2}) & /@ Range[51, 68];
```

In[∘]:=
```
Images51to68 = PolyPlot2 /@ Data51to68;
```

In[∘]:=
```
Show[Beehive = Graphics[{
    {0, 0, 1}, {1, 0, 2}, {2, 0, 3},
    {0, 1, 4}, {1, 1, 5},
    {-1, 2, 6}, {0, 2, 7}, {1, 2, 8},
    {-1, 3, 9}, {0, 3, 10},
    {-2, 4, 11}, {-1, 4, 12}, {0, 4, 13},
    {-2, 5, 14}, {-1, 5, 15},
    {-3, 6, 16}, {-2, 6, 17}, {-1, 6, 18}
   },
   PlotRange → {{-0.6, 6.1}, {0, 2}}, ImageSize → 1200] /. {x_, y_, k_} ⧴
   Inset[
    SetAlphaChannel[Images51to68⟦k⟧, Graphics[{White,
       Polygon[Table[{Cos[t], Sin[t]}, {t, 0, 5 π / 3, π / 3}]]}, Background → Black]],
    x {0, 1} + y {√3 / 2, 1 / 2}, Center, 1.2
   ], ImageSize → Small]
```

Out[∘]=



In[∘]:=
```
Export["figs/Beehive.pdf", ImageCrop[Beehive]]
```

Out[∘]=

figs/Beehive.pdf

Θ Program

Program

```
CF[ℰ_] := Expand@Collect[ℰ, g__, F] /. F → Factor;
T3 = T1 T2;

F1[{s_, i_, j_}] := CF[
  s (1/2 - g3ii + T2^s g1ii g2ji - g1ii g2jj - (T2^s - 1) g2ji g3ii + 2 g2jj g3ii - (1 - T3^s) g2ji g3ji -
     g2ii g3jj - T2^s g2ji g3jj + g1ii g3jj + ((T1^s - 1) g1ji (T2^(2 s) g2ji - T2^s g2jj + T2^s g3jj) +
        (T3^s - 1) g3ji (1 - T2^s g1ii + g2ij + (T2^s - 2) g2jj - (T1^s - 1) (T2^s + 1) g1ji)) / (T2^s - 1))]
F2[{s0_, i0_, j0_}, {s1_, i1_, j1_}] := CF[s1 (T1^s0 - 1) (T2^s1 - 1)^-1
   (T3^s1 - 1) g1,j1,i0 g3,j0,i1 ((T2^s0 g2,i1,i0 - g2,i1,j0) - (T2^s0 g2,j1,i0 - g2,j1,j0))]
F3[φ_, k_] = φ g3kk - φ / 2;

Θ[K_] := Θ[K] = Module[{X, φ, n, A, Δ, G, ev, θ, k, k1, k2},
    {X, φ} = Rot[K]; n = Length[X]; A = IdentityMatrix[2 n + 1];
    Cases[X, {s_, i_, j_} ⧴ (A[[{i, j}, {i + 1, j + 1}]] += ( -T^s  T^s - 1
                                                              0      -1   ))];
    Δ = T^((-Total[φ] - Total[X[[All, 1]]]) / 2) Det[A];
    G = Inverse[A];
    ev[ℰ_] := Factor[ℰ /. g_{ν_, α_, β_} ⧴ (G[[α, β]] /. T → T_ν)];
    θ = ev[Sum[F1[X[[k]]], {k, n}]];
    θ += ev[Sum[F2[X[[k1]], X[[k2]]], {k1, n}, {k2, n}]];
    θ += ev[Sum[F3[φ[[k]], k], {k, Length@φ}]];
    Factor@{Δ, (Δ /. T → T1) (Δ /. T → T2) (Δ /. T → T3) θ}
   ];
```

In[∘]:=

```
Expand[Θ[Knot[3, 1]]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

Out[∘]=

$$\left\{-1+\frac{1}{T}+T,\ -\frac{1}{T_1^2}-T_1^2-\frac{1}{T_2^2}-\frac{1}{T_1^2 T_2^2}+\frac{1}{T_1 T_2^2}+\frac{1}{T_1^2 T_2}+\frac{T_1}{T_2}+\frac{T_2}{T_1}+T_1^2 T_2-T_2^2+T_1 T_2^2-T_1^2 T_2^2\right\}$$

In[∘]:=

```
Export[
 "figs/Program.pdf",
 Append[
  First@Cases[NotebookGet@EvaluationNotebook[], Cell[__, CellTags → "Program", ___], ∞],
  PageWidth → 600
 ]
]
```

Out[∘]=

figs/Program.pdf

In[$\circ$]:= `Export["figs/ThetaConway.pdf", Echo@PolyPlot2[Θ[Knot@"K11n34"][[2]]]]`

»



Out[$\circ$]=

figs/ThetaConway.pdf

In[ ]:=
```
Export["figs/ThetaKT.pdf", Echo@PolyPlot2[Θ[Knot@"K11n42"]〚2〛]]
```

»


Out[ ]=
```
figs/ThetaKT.pdf
```

The Rolfsen Table

In[ ]:=
```
tab250 = Join[
   {{1, 0}},
   Table[Θ[K], {K, AllKnots[{3, 10}]}]
  ];
```

In[ ]:=
```
g250 = GraphicsGrid[
  Partition[PadRight[PolyPlot /@ tab250, 256, Graphics[]], 16],
  Spacings → 0, Dividers → All, ImagePadding → None, PlotRangePadding → None
 ]
```

Out[ ]=



In[ ]:=
```
Export["figs/Theta4Rolfsen.pdf", g250]
```

Out[ ]=

figs/Theta4Rolfsen.pdf

Pretzel Knots

In[◦]:=
```
(*L should be a list of an odd number of odd integers.*)
OddPretzel[L_] := Module[{s = Total[Abs /@ L], x},
  PD @@ Flatten@Table[x = Sum[Abs[L[[j]]], {j, 1, u - 1}];

     If[EvenQ[u],
      If[L[[u]] > 0,
       Join[Table[X[s + x + Abs[L[[u]]] - 2 k, x + 2 + 2 k,
           s + x + Abs[L[[u]]] - 2 k - 1, x + 1 + 2 k], {k, 0, (Abs[L[[u]]] - 1)/2}],
        Table[X[x + 2 k, s + x + Abs[L[[u]]] + 2 - 2 k,
           x + 1 + 2 k, s + x + Abs[L[[u]]] + 1 - 2 k], {k, 1, (Abs[L[[u]]] - 1)/2}]],
       Join[Table[X[x + 1 + 2 k, s + x + Abs[L[[u]]] - 2 k,
           x + 2 + 2 k, s + x + Abs[L[[u]]] + 1 - 2 k], {k, 0, (Abs[L[[u]]] - 1)/2}],
        Table[X[s + x + Abs[L[[u]]] + 1 - 2 k, x + 2 k,
           s + x + Abs[L[[u]]] + 2 - 2 k, x + 1 + 2 k], {k, 1, (Abs[L[[u]]] - 1)/2}]]]
      ,
      If[L[[u]] > 0,
       Join[Table[X[x + 1 + 2 k, s + x + Abs[L[[u]]] + 1 - 2 k,
           x + 2 + 2 k, s + x + Abs[L[[u]]] - 2 k], {k, 0, (Abs[L[[u]]] - 1)/2}],
        Table[X[s + x + Abs[L[[u]]] + 1 - 2 k, x + 1 + 2 k,
           s + x + Abs[L[[u]]] + 2 - 2 k, x + 2 k], {k, 1, (Abs[L[[u]]] - 1)/2}]],
       Join[Table[X[s + x + Abs[L[[u]]] - 2 k, x + 1 + 2 k,
           s + x + Abs[L[[u]]] + 1 - 2 k, x + 2 + 2 k], {k, 0, (Abs[L[[u]]] - 1)/2}],
        Table[X[x + 2 k, s + x + Abs[L[[u]]] + 1 - 2 k,
           x + 1 + 2 k, s + x + Abs[L[[u]]] + 2 - 2 k], {k, 1, (Abs[L[[u]]] - 1)/2}]]]
      ]
     , {u, Length[L]}]
  ]
```

In[◦]:=
```
(*L should be a list L={L0,L1,..,Ln} where L0 and n≥0 are even and the Li,
i>0 are odd integers.*)
```

```
EOddPretzel[LL_] := Module[{L = Rest@LL, s, x, y, n},
  s = 2 Abs[LL[[1]]] + Total[Abs /@ L]; n = Length[L];
  PD @@ Flatten@
    Join[
     If[LL[[1]] > 0,
      Join[Table[X[1 + 2 k, s + 1 - 2 k, 2 + 2 k, s - 2 k], {k, 0, Abs[LL[[1]]]/2 - 1}],
       Table[X[s + 1 - 2 k, 1 + 2 k, s + 2 - 2 k, 2 k], {k, 1, Abs[LL[[1]]]/2}]],
      Join[Table[X[s - 2 k, 1 + 2 k, s + 1 - 2 k, 2 + 2 k], {k, 0, Abs[LL[[1]]]/2 - 1}],
       Table[X[2 k, s + 1 - 2 k, 1 + 2 k, s + 2 - 2 k], {k, 1, Abs[LL[[1]]]/2}]]
     ],
     Table[x = Sum[Abs[L[[j]]], {j, u + 1, n}];
      y = Abs[LL[[1]]] + Sum[Abs[L[[j]]], {j, 1, u - 1}];

      If[EvenQ[u],
       If[L[[u]] > 0,
        Join[
         Table[X[s + x + 1 + 2 k, y + 1 + 2 k, s + x + 2 + 2 k, y + 2 + 2 k], {k, 0, (Abs[L[[u]]] - 1)/2}],
         Table[X[y + 2 k, s + x + 2 k, y + 1 + 2 k, s + x + 1 + 2 k], {k, 1, (Abs[L[[u]]] - 1)/2}]],
        Join[
         Table[X[y + 1 + 2 k, s + x + 2 + 2 k, y + 2 + 2 k, s + x + 1 + 2 k], {k, 0, (Abs[L[[u]]] - 1)/2}],
         Table[X[s + x + 2 k, y + 1 + 2 k, s + x + 1 + 2 k, y + 2 k], {k, 1, (Abs[L[[u]]] - 1)/2}]]]
       ,
       If[L[[u]] > 0,
        Join[
         Table[X[y + 1 + 2 k, s + x + 1 + 2 k, y + 2 + 2 k, s + x + 2 + 2 k], {k, 0, (Abs[L[[u]]] - 1)/2}],
         Table[X[s + x + 2 k, y + 2 k, s + x + 1 + 2 k, y + 1 + 2 k], {k, 1, (Abs[L[[u]]] - 1)/2}]],
        Join[
         Table[X[s + x + 1 + 2 k, y + 2 + 2 k, s + x + 2 + 2 k, y + 1 + 2 k], {k, 0, (Abs[L[[u]]] - 1)/2}],
```

```
            Table[X[y + 2 k, s + x + 1 + 2 k, y + 1 + 2 k, s + x + 2 k], {k, 1, (Abs[L⟦u⟧] - 1)/2}]]]
        ]
        , {u, Length[L]}]
      ]
  ]
```

In[∘]:= `Crossings@EOddPretzel[{2, 41, -41}]`

Out[∘]=

84

In[∘]:= `P41 = PolyPlot2[Θ[EOddPretzel[{2, 41, -41}]]⟦2⟧]`

Out[∘]=



In[∘]:= `Export["figs/P41.pdf", P41]`

Out[∘]=

figs/P41.pdf